

複素数と Mathematica

桂田 祐史

2015年9月28日, 2023年6月26日

<https://m-katsurada.sakura.ne.jp/complex/mathematica-memo/>
<http://nalab.mind.meiji.ac.jp/~mk/complex/mathematica-memo/>

色々なことをコンピューターが計算してくれる時代になった。私は、研究上で現れる問題はもちろん、授業中の計算例とか試験問題の解答の確認をするためにも、日常的にコンピューターを使っている。ゼミでも積極的にコンピューターを活用するように言っている。期末試験などでは、コンピューター持ち込み禁止としてあって、それは急には変えられないだろうが、宿題等では活用するように言っている。

(2021年度は、COVID19の流行のため、期末試験がレポートに置き換わって、コンピューター利用が役立つ状況になったはずだが、活用しなかった人が多かったのは残念だった。)

実際にはコンピューターで計算するにしてもコンピューターを使いこなすには、きちんと使いこなすための準備が必要である。

本音を言うと、必要性をこちらが説く前に、自分から興味を持って取り組んでもらいたいところ。

Mathematica のバージョンによって挙動が違うことがあるので、バージョンを添えて結果も載せた方が良いのかもしれない。

(2020/1/19) 学生で「Wolfram Alpha¹ でやっています」という人もいます。確かに、ここに載っていることくらいは、Wolfram Alpha でもやってくれるみたいだ。

1 全般的な覚え書き

- 現象数理学科でライセンスを購入しているので、所属する学生は利用できる。毎年4月末日にライセンスの更新がある(自動更新されるはずであるが、何か問題が生じて使えなくなっていた場合は、桂田か池田先生に相談する)。
- アプリケーション・フォルダに Mathematica.app がある(私は Dock に追加しています)。そこからならほぼ確実に起動できる。
- (新しくプログラムを作る場合) Mathematica を起動後、「新規ドキュメント」でノートブックを開き、コマンドを入力して実行する。
- 忘れないように: コマンドの最後に `shift`+`↵` とタイプする。
- 直前の結果は % で参照できる。直前のコマンドは `command`+L で呼び出せる。

¹<https://www.wolframalpha.com/>

- コマンドは編集して再実行できる (挿入、上書き修正、削除、などが可能)。
- ??関数名 としてマニュアル (Documentation) が開ける (非常に便利。これに慣れること。)
- 関数名の大文字・小文字に注意する。ほぼ例外なく、先頭は大文字である。
- ノートブックとして保存しておける (ファイル名末尾 .nb)。
- 既存のノートブックはダブルクリックで開ける。コマンドを1つ1つ `shift`+`↵` で実行する以外に、`[評価]` → `[ノートブックを評価]` で順番に全部実行することもできる。
- 「[Mathematica 入門](#)」

2 四則など簡単な演算

虚数単位は I (大文字) で表す。絶対値 (absolute value) は `Abs[]`, 偏角 (argument) の主値は `Arg[]`, 共役複素数 (complex conjugate) は `Conjugate[]` で計算できる。

2005年度問1² (1) の検算に使ってみる。

```
a=1+I
b=2+3I
a+b
a-b
a b
a/b
Abs[a]
Conjugate[a]
Arg[a]
a^4
```

注意 `a b` は `a*b` (a と b の積) を意味する。空白を省略して `ab` とすると、1つの名前になってしまう。

²<http://nalab.mind.meiji.ac.jp/~mk/lecture/complex-function-2015/toi1.pdf>

名称未定義-1 100%

In[3]:= **a + b**

Out[3]= $3 + 4 i$

In[4]:= **a - b**

Out[4]= $-1 - 2 i$

In[5]:= **a b**

Out[5]= $-1 + 5 i$

In[6]:= **a / b**

Out[6]= $\frac{5}{13} - \frac{i}{13}$

In[7]:= **Abs[a]**
[絶対値]

Out[7]= $\sqrt{2}$

In[8]:= **Conjugate[a]**
[複素共役]

Out[8]= $1 - i$

In[9]:= **Arg[a]**
[偏角]

Out[9]= $\frac{\pi}{4}$

In[10]:= **a ^ 4**

Out[10]= -4

絶対値 名前 ▾ ↺ ⚙️ 💬

+

3 平方根の計算

$c = a + ib$ ($a, b \in \mathbb{R}$) が与えられた時に、 $z^2 = c$ の解を $z = x + iy$ ($x, y \in \mathbb{R}$) の形で求めることが出来る。

(複素数の平方根が、実数の $\sqrt{\quad}$ で表現できる、という定理に基づく。)

$(x + iy)^2 = a + ib$ より連立方程式

$$x^2 - y^2 = a, \quad 2xy = b$$

が得られるので、その実数の範囲の解を求めれば良い。

$1 + i$ の平方根を求めてみよう。

```
a=1
b=1
sol=Solve[{x^2-y^2==a,2 x y==b},{x,y},Reals]
FullSimplify[sol]

あるいは ToRadicals[sol]
```

以上は授業で説明したやり方に沿って Mathematica に仕事をさせるものだが、 z の方程式のまま解かせることも出来る (Mathematica が内部で何をしているのかは謎だけど)。

```
sol=Solve[z^2 == 1 + I, z]
sol2=ComplexExpand[sol]
ToRadicals[sol2]
```

最初に $z^2 = 1 + i$ を解かせると $z = \pm\sqrt{1+i}$ となるが、ComplexExpand[] で実部・虚部に展開させると、 $z = \pm\sqrt[4]{2} \left(\cos \frac{\pi}{8} + i \sin \frac{\pi}{8} \right)$ となり、ToRadicals[] で処理すると、 $z = \pm \left(\frac{\sqrt{2+\sqrt{2}}}{2^{3/4}} + i \frac{\sqrt{2-\sqrt{2}}}{2^{3/4}} \right)$ となる。

名称未定義-2 100%

In[11]:= **a = 1**

Out[11]= 1

In[12]:= **b = 1**

Out[12]= 1

In[13]:= **sol = Solve[{x^2 - y^2 == a, 2 x y == b}, {x, y}, Reals]**
解く 実数領域

Out[13]= $\left\{ \left\{ x \rightarrow -1.10\dots, y \rightarrow -2 \sqrt{-1.10\dots + 2 \sqrt{-1.10\dots^3}} \right\}, \left\{ x \rightarrow 1.10\dots, y \rightarrow -2 \sqrt{1.10\dots + 2 \sqrt{1.10\dots^3}} \right\} \right\}$

In[14]:= **ToRadicals[sol]**
根基で

Out[14]= $\left\{ \left\{ x \rightarrow -\sqrt{\frac{1}{2} + \frac{1}{\sqrt{2}}}, y \rightarrow 2 \sqrt{\frac{1}{2} + \frac{1}{\sqrt{2}}} - 2 \left(\frac{1}{2} + \frac{1}{\sqrt{2}} \right)^{3/2} \right\}, \left\{ x \rightarrow \sqrt{\frac{1}{2} + \frac{1}{\sqrt{2}}}, y \rightarrow -2 \sqrt{\frac{1}{2} + \frac{1}{\sqrt{2}}} + 2 \left(\frac{1}{2} + \frac{1}{\sqrt{2}} \right)^{3/2} \right\} \right\}$

In[15]:= **FullSimplify[sol]**
完全に簡約

Out[15]= $\left\{ \left\{ x \rightarrow -\sqrt{\frac{1}{2} + \frac{1}{\sqrt{2}}}, y \rightarrow -\sqrt{-\frac{1}{2} + \frac{1}{\sqrt{2}}} \right\}, \left\{ x \rightarrow \sqrt{\frac{1}{2} + \frac{1}{\sqrt{2}}}, y \rightarrow \sqrt{-\frac{1}{2} + \frac{1}{\sqrt{2}}} \right\} \right\}$

In[16]:= **sol = Solve[z^2 == 1 + I, z]**
解く 虚数単位

Out[16]= $\left\{ \left\{ z \rightarrow -\sqrt{1 + i} \right\}, \left\{ z \rightarrow \sqrt{1 + i} \right\} \right\}$

In[17]:= **sol2 = ComplexExpand[sol]**
式の展開

Out[17]= $\left\{ \left\{ z \rightarrow -2^{1/4} \cos\left[\frac{\pi}{8}\right] - i 2^{1/4} \sin\left[\frac{\pi}{8}\right] \right\}, \left\{ z \rightarrow 2^{1/4} \cos\left[\frac{\pi}{8}\right] + i 2^{1/4} \sin\left[\frac{\pi}{8}\right] \right\} \right\}$

In[18]:= **ToRadicals[sol2]**
根基で

Out[18]= $\left\{ \left\{ z \rightarrow -\frac{i \sqrt{2 - \sqrt{2}}}{2^{3/4}} - \frac{\sqrt{2 + \sqrt{2}}}{2^{3/4}} \right\}, \left\{ z \rightarrow \frac{i \sqrt{2 - \sqrt{2}}}{2^{3/4}} + \frac{\sqrt{2 + \sqrt{2}}}{2^{3/4}} \right\} \right\}$

4 実部虚部への分解

$\Omega \subset \mathbb{C}$, $f: \Omega \rightarrow \mathbb{C}$ とするとき、


$$u(x, y) := \operatorname{Re} f(x + iy), \quad v(x, y) := \operatorname{Im} f(x + iy) \quad ((x, y) \in \tilde{\Omega} := \{(x, y) \in \mathbb{R}^2 \mid x + iy \in \Omega\})$$

で定めた u, v (複素関数の実部・虚部) が必要になる場合がある。

これらを求めるには、既に紹介した `ComplexExpand[]` を用いると良い。

```
f[z_]:=z^3
ComplexExpand[f[x+I y]]

g[z_]:=Cos[z]
ComplexExpand[g[x+I y]]
```



```
sec3.nb 100%
In[34]:= f[z_] := z^3
In[35]:= ComplexExpand[f[x + I y]]
Out[35]= x^3 - 3 x y^2 + i (3 x^2 y - y^3)

In[36]:= g[z_] := Cos[z]
Out[36]= Cos[x] Cosh[y] - i Sin[x] Sinh[y]

In[38]:= (* 以下はおまけ *)
In[39]:= u[x_, y_] = ComplexExpand[Re[f[x + I y]]]
Out[39]= x^3 - 3 x y^2

In[40]:= v[x_, y_] = ComplexExpand[Im[f[x + I y]]]
Out[40]= 3 x^2 y - y^3

In[41]:= u[x, y]
Out[41]= x^3 - 3 x y^2

In[42]:= v[x, y]
Out[42]= 3 x^2 y - y^3

In[43]:= D[{u[x, y], v[x, y]}, {{x, y}}]
Out[43]= {{3 x^2 - 3 y^2, -6 x y}, {6 x y, 3 x^2 - 3 y^2}}
```

以下を想定しています: 行列 | 代りに ベアのリスト として使う

表示形式 ▼ 逆 行列式 固有値 ▾ その他... ↻ ⚙


5 Taylor 展開, Laurent 展開

SeriesCoefficient[式, {z,c,n}] で、 c のまわりの冪級数展開 $\sum_{n=0}^{\infty} a_n(z-c)^n$ や、 c のまわりの Laurent 級数展開 $\sum_{n=-\infty}^{\infty} a_n(z-c)^n$ の第 n 項の係数 a_n を表示してくれる (とても便利)。

Series[式, {変数, c, n}] で、 c のまわりの Taylor 展開を第 n 項まで計算出来る。

```
SeriesCoefficient[Cos[z], {z, 0, n}]
SeriesCoefficient[Tan[z], {z, 0, n}]
SeriesCoefficient[Cot[z], {z, 0, n}]
SeriesCoefficient[1/(2+z), {z, 0, n}]
```

```
Series[Cos[z], {z, 0, 10}]
Series[1/(2+z), {z, 0, 10}]
```



The screenshot shows a Mathematica notebook window titled "名称未定義-4" with a zoom level of 100%. It displays the following input and output pairs:

- In[44]:** SeriesCoefficient[Cos[z], {z, 0, n}]
級数の係数 余弦
Out[44]:
$$\begin{cases} \frac{i^n (1 + (-1)^n)}{2^n n!} & n \geq 0 \\ 0 & \text{True} \end{cases}$$
- In[45]:** SeriesCoefficient[Tan[z], {z, 0, n}]
級数の係数 正接
Out[45]:
$$\begin{cases} \frac{i^{1+n} 2^n (-1 + (-1)^n) (-1 + 2^{1+n}) \text{BernoulliB}[1+n]}{(1+n)!} & n \geq 1 \\ 0 & \text{True} \end{cases}$$
- In[46]:** SeriesCoefficient[Cot[z], {z, 0, n}]
級数の係数 余接
Out[46]:
$$\begin{cases} 1 & n == -1 \\ -\frac{i (2 i)^n (-1 + (-1)^n) \text{BernoulliB}[1+n]}{(1+n)!} & n \geq 0 \\ 0 & \text{True} \end{cases}$$
- In[47]:** SeriesCoefficient[1/(2+z), {z, 0, n}]
級数の係数
Out[47]:
$$\begin{cases} (-1)^n 2^{-1-n} & n \geq 0 \\ 0 & \text{True} \end{cases}$$
- In[48]:** Series[Cos[z], {z, 0, 10}]
級数展開 余弦
Out[48]:
$$1 - \frac{z^2}{2} + \frac{z^4}{24} - \frac{z^6}{720} + \frac{z^8}{40320} - \frac{z^{10}}{3628800} + O[z]^{11}$$
- In[49]:** Series[1/(2+z), {z, 0, 10}]
級数展開
Out[49]:
$$\frac{1}{2} - \frac{z}{4} + \frac{z^2}{8} - \frac{z^3}{16} + \frac{z^4}{32} - \frac{z^5}{64} + \frac{z^6}{128} - \frac{z^7}{256} + \frac{z^8}{512} - \frac{z^9}{1024} + \frac{z^{10}}{2048} + O[z]^{11}$$

At the bottom of the notebook, there is a toolbar with icons for "高次の項を切断" (Cut off higher-order terms), "係数のリスト" (List of coefficients), "n次の係数..." (Coefficient of order n...), "逆級数" (Inverse series), "パデ近似" (Padé approximation), and a settings icon.

6 級数の和

Mathematica の `Sum[]` は級数の和を計算してくれる。思いの外、強力である。

```
Sum[(-1)^n z^(2n)/(2n)!, {n, 0, Infinity}]
```

に対して `Cos[z]` という答を返す。

Taylor 展開、Laurent 展開を計算したとき、その結果を `Sum[]` に与えることで検算が可能である。

The screenshot shows the Mathematica interface with the following content:

```
In[51]:= Sum[(-1)^n z^(2n)/(2n)!, {n, 0, Infinity}]
Out[51]= Cos[z]
```

The second calculation is partially visible:

```
In[53]:= Sum[(-1)^n z^(-1-n), {n, 0, Infinity}]
Out[53]= 1/(2+z)
```

The interface includes a toolbar with buttons for 'プロット' (Plot), '分母' (Denominator), 'zについて微分' (Differentiate with respect to z), 'zについて積分' (Integrate with respect to z), and 'その他...' (Other...).

7 部分分数分解

有理式の部分分数への分解が必要になる場合があるが、`Apart[]` で計算できる。

```
Apart[(z^3-3z^2-z+5)/(z^2-5z+6)]
```

The screenshot shows a Mathematica notebook window titled "名称未定義-6" at 100% zoom. The input cell contains the command `Apart[(z^3 - 3z^2 - z + 5) / (z^2 - 5z + 6)]` with a tooltip indicating it performs partial fraction decomposition of a rational function. The output cell displays the result: $2 + \frac{2}{-3 + z} + \frac{1}{-2 + z} + z$. Below the output is a toolbar with buttons for "プロット", "通分して約分", "zについて微分", "zについて積分", and "その他...", along with icons for undo, settings, and help.

8 応用: ある計算問題 (有理関数の Taylor 展開を求める) の答の検算

$$f(z) = \frac{z^3 - 3z^2 - z + 5}{z^2 - 5z + 6}.$$

の 0 の周りの Taylor 展開を求めよ、という問題。

人手で解く場合は、 $f(z)$ を部分分数に分解する。そのためには、 $z^3 - 3z^2 - z + 5$ を $z^2 - 5z + 6$ で割りたくなる。

```
A=z^3-3z^2-z+5
```

```
B=z^2-5z+6
```

```
f=A/B
```

```
q=PolynomialQuotient[A,B,z]
```

```
r=PolynomialRemainder[A,B,z]
```

あるいは

```
{q,r}=PolynomialQuotientRemainder[A,B,z]
```

これから商 $q = z + 2$, 余り $r = 3z - 7$ が求まる (商は quotient, 余りは remainder. polynomial は多項式という意味)。ゆえに

$$f(z) = \frac{(z+2)(z^2-5z+6) + 3z-7}{z^2-5z+6} = z+2 + \frac{3z-7}{(z-2)(z-3)}.$$

この右辺を部分分数分解しても良いが、そもそも Mathematica にやらせるのならば、最初から $f(z)$ の部分分数分解を指示してもよい。

```
Apart[r/B]
```

```
Apart[f]
```

それぞれ $\frac{2}{-3+z} + \frac{1}{-2+z}$, $2 + \frac{2}{-3+z} + \frac{1}{-2+z} + z$ となる。

結局

$$f(z) = z + 2 + \frac{1}{z-2} + \frac{2}{z-3}.$$

`Series[f, {z, 0, 10}]` とすると、0 の周りの Taylor 展開を 10 次の項まで求めることが出来る。

$$f(z) = \frac{5}{6} + \frac{19z}{36} - \frac{43}{216}z^2 - \frac{113}{1296}z^3 - \frac{307}{7776}z^4 - \dots (\text{途中省略}) - \dots - \frac{181243}{362797056}z^{10} + O(z^{11})$$

が得られる。series は級数という意味の英単語である。

実は、0 の周りの Taylor 展開の第 n 項は

```
SeriesCoefficient[f, {z, 0, n}]
```

で求められる (宿題の答えだけを求めるには、これ一発で OK)。

```

名称未定義-7 100%
In[55]:= A = z ^ 3 - 3 z ^ 2 - z + 5
Out[55]= 5 - z - 3 z ^ 2 + z ^ 3

In[56]:= B = z ^ 2 - 5 z + 6
Out[56]= 6 - 5 z + z ^ 2

In[57]:= f = A / B
Out[57]= (5 - z - 3 z ^ 2 + z ^ 3) / (6 - 5 z + z ^ 2)

In[58]:= q = PolynomialQuotient[A, B, z]
Out[58]= 2 + z

In[59]:= r = PolynomialRemainder[A, B, z]
Out[59]= -7 + 3 z

In[60]:= Apart[r / B]
Out[60]= 2 / (-3 + z) + 1 / (-2 + z)

In[61]:= Apart[f]
Out[61]= 2 + 2 / (-3 + z) + 1 / (-2 + z) + z

In[62]:= SeriesCoefficient[f, {z, 0, n}]
Out[62]= { 19/36, 5/6, -6^-1-n (2^2+n + 3^1+n), 0 }
           n == 1
           n == 0
           n > 1
           True

```

f の 0 の周りの Taylor 展開は

$$f(z) = \frac{5}{6} + \frac{19}{36}z - \sum_{n=2}^{\infty} \left(\frac{1}{2^{n+1}} + \frac{2}{3^{n+1}} \right) z^n.$$

Sum[] で検算が可能で、

```

5/6+19z/36-Sum[(1/2^(n+1)+2/3^(n+1))z^n,{n,2,Infinity}]
Simplify[%]

```

とすると、

$$\frac{5 - z - 3z^2 + z^3}{6 - 5z + z^2}$$

が得られる。無事、 $f(z)$ と一致したので、ほっと一息。

9 複素対数関数を描く

2変数 (x, y) の関数としての $\text{Im Log}(x + iy)$, $\text{Re Log}(x + iy)$ のグラフを描いてみよう。

それぞれ $\text{Arg}(x + iy)$, $\log \sqrt{x^2 + y^2}$ であるから、コンピューターで図示しなくても分からなくはないが (図示しなくても分かるけれど)、やってみることを勧める。

`Plot3D[]` や `ContourPlot[]` では、描画範囲を x 座標と y 座標の範囲で指定するので、変数は `x+I y` と書くと良い (小さなノウハウ)。

Mathematica でグラフを描こう

```
Plot3D[Im[Log[x+I y]], {x,-2,2}, {y,-2,2},
```

```
  RegionFunction->Function[{x,y,z},x^2+y^2<4]]
```

```
Plot3D[Re[Log[x+I y]], {x,-2,2}, {y,-2,2},
```

```
  RegionFunction->Function[{x,y,z},x^2+y^2<4]]
```

`RegionFunction[]` は $x^2 + y^2 < 4$ の範囲だけでグラフを描くための指定 (なくても描けるし、絞るのは趣味の問題)。




図 1: $\text{Im Log}(x + yi)$, $\text{Re Log}(x + yi)$ のグラフ


(以前は描画範囲を絞るため、`RegionFunction` を使うのではなく、`Re[Log[x+I y]]Boole[x^2+y^2<4]` のグラフを描いていた。)

Mathematica で描いたグラフは、マウスでつかんでグリグリ動かせる。ぜひやってみること (静止画を見るだけだと今ひとつ分かりにくい)。

`Plot3D[]` の代わりに `ContourPlot[]` を用いると、レベル表示 (≡等高線描画) 出来る。

```
ContourPlot[Im[Log[x+I y]], {x,-2,2}, {y,-2,2}, Contours->Table[x,{x,-Pi,Pi,Pi/8}]]
```

```
ContourPlot[Re[Log[x+I y]], {x,-2,2}, {y,-2,2}]
```



10 Abel の連続性定理に現れる収束範囲

冪級数 $\sum_{n=0}^{\infty} a_n(z-c)^n$ が収束円周上のある点 z_0 で収束するならば、その冪級数は “Stolz の角領域” で一様収束するので、和はそこで連続な関数である、というのが Abel の連続性定理で、それにより、

$$\frac{\pi}{4} = 1 - \frac{1}{3} + \frac{1}{5} - \dots + \frac{(-1)^{n-1}}{2n-1} + \dots$$

などの有名な結果が証明できる。

例えば $c = 0, z_0 = R$ の場合、

$$\Omega_K := \left\{ z \in \mathbb{C} \mid |z| < R, \frac{|1 - z/R|}{1 - |z|/R} < K \right\}$$

であるが、これは一体どういう形なのだろうか？

```

stolz[K_, R_] :=
Block[{g1, g2},
g1 = ContourPlot[x^2 + y^2 == R^2, {x, -2 R, 2 R}, {y, -2 R, 2 R}];
g2 = RegionPlot[
x^2 + y^2 < R^2 &&
Abs[1 - (x + I y)/R]/(1 - Abs[x + I y]/R) <= K, {x, -2 R,
2 R}, {y, -2 R, 2 R}]; Show[g1, g2]
]

R=1
Manipulate[stolz[K,R],{K,1,10,0.2}]

```

筆者は、Mathematica を使うまで、 Ω_K がどういう形をしているか、実は良く分かっていなかった(そんなに難しくもないけれど、ちょっと考えて分かるものでもなくて、何となく気になってはいたけれど、放置していました。)

11 線積分の験算

以下の線積分の値を求めよ。


(1) $C: z = t + it^2$ ($t \in [0, 1]$) とするとき $I_1 = \int_C \operatorname{Re} z \, dz$ (2) $c \in \mathbb{C}, r > 0, n \in \mathbb{N}$,

$C: z = c + re^{i\theta}$ ($\theta \in [0, 2\pi]$) とするとき $I_2 = \int_C \frac{dz}{(z-c)^n}$ (3) 0 から $1+i$ に至る線分を C

とするとき $I_3 = \int_C \operatorname{Im} z \, dz$ (4) 単位円 $|z| = 1$ の下半分を -1 から 1 までたどる曲線を C

とするとき $I_4 = \int_C \bar{z} \, dz$ (5) 図の正方形の周を反時計回りに一周する曲線を C とするとき

$I_5 = \int_C |z| \, dz, I_6 = \int_C (z^2 + 3z + 4) \, dz$






図 2: K を大きくすると膨れていきます

```

z=t+I t^2;
I1=Integrate[Re[z]D[z,t],{t,0,1}]

z=c+r Exp[I t];
I2=Integrate[1/(z-c) D[z,t],{t,0,2Pi}]
I2a=Integrate[1/(z-c)^n D[z,t],{t,0,2Pi}]

z=(1+I)t;
I3=Integrate[Im[z] D[z,t],{t,0,1}]

z=Exp[I t];
I4=Integrate[Conjugate[z] D[z,t],{t,Pi,2Pi}]

z1=t;
z2=1+I*t;
z3=1+I-t;
z4=I-I*t;
I5=Integrate[Abs[z1]D[z1,t],{t,0,1}]+Integrate[Abs[z2]D[z2,t],{t,0,1}]
+Integrate[Abs[z3]D[z1,t],{t,0,1}]+Integrate[Abs[z4]D[z2,t],{t,0,1}]


f[z_] := z^2 + 3 z + 4
I6 = Integrate[f[z1] D[z1, t], {t, 0, 1}] +
Integrate[f[z2] D[z2, t], {t, 0, 1}] +
Integrate[f[z3] D[z3, t], {t, 0, 1}] +
Integrate[f[z4] D[z4, t], {t, 0, 1}]

```

答は (1) $I_1 = \frac{1}{2} + \frac{2}{3}i$ (2) $n = 1$ のとき $I_2 = 2\pi i$, $n \neq 1$ のとき $I_2 = 0$ (3) $I_3 = \frac{1+i}{2}$ (4) $I_4 = \pi i$ (5) $I_5 = \frac{i-1}{2} (\sqrt{2}-1 + \log(1+\sqrt{2}))$ (6) $I_6 = 0$

Mathematica は、(5) の $\log(1+\sqrt{2})$ を $\text{ArcSinh}[1]$ と表示する。 $\text{TrigToExp}[]$ を施すと \log で表示してくれる。

```
In[ ]:= TrigToExp[ArcSinh[1]]
Out[ ]= Log[1+√2]
```




```

名称未定義-10 100%
In[145]:= z1 = t;
z2 = 1 + I * t;
z3 = 1 + I - t;
z4 = I - I * t;
I5 = Integrate[Abs[z1] D[z1, t], {t, 0, 1}] + Integrate[Abs[z2] D[z2, t], {t, 0, 1}] +
Integrate[Abs[z3] D[z3, t], {t, 0, 1}] + Integrate[Abs[z4] D[z4, t], {t, 0, 1}]
Out[149]=  $\left(\frac{1}{2} + \frac{i}{2}\right) + \left(\frac{1}{2} + \frac{i}{2}\right) (\sqrt{2} + \text{ArcSinh}[1])$ 
In[153]:= TrigToExp[I5]
Out[153]=  $\left(\frac{1}{2} + \frac{i}{2}\right) + \frac{1+i}{\sqrt{2}} + \left(\frac{1}{2} + \frac{i}{2}\right) \text{Log}[1 + \sqrt{2}]$ 
In[155]:= Clear[f]
In[156]:= f[z_] := z^2 + 3 z + 4
In[157]:= I6 = Integrate[f[z1] * D[z1, t], {t, 0, 1}] + Integrate[f[z2] * D[z2, t], {t, 0, 1}] +
Integrate[f[z3] * D[z3, t], {t, 0, 1}] + Integrate[f[z4] * D[z4, t], {t, 0, 1}]
Out[157]= 0

```

12 曲線の連続的な変形 (ホモトピー)

円を楕円に変形する。

```
phi0[t_]:= {Cos[t], Sin[t]};  
  
phi1[t_]:= {3Cos[t], 2Sin[t]};  
  
F[t_, u_] := (1-u)phi0[t] + u*phi1[t];  
  
Manipulate[ParametricPlot[{phi0[t], phi1[t], F[t, u]}, {t, 0, 2 Pi},  
  PlotRange -> {{-3, 3}, {-3, 3}}, {u, 0, 1}]
```

$\phi_1[t] := \{1/2, 0\}$ とすると、定数曲線で、像は $\{(1/2, 0)\}$ であり、円周 $x^2 + y^2 = 1$ を 1 点 $(1/2, 0)$ に変形することになる。

以上は複素数を使っていないので、書き換えてみる。

```
Clear[phi0, phi1, F]  
phi0[t_] := Exp[I t];  
phi1[t_] := 3Cos[t] + 2 I Sin[t];  
F[t_, u_] := (1-u)phi0[t] + u*phi1[t];  
z2xy[z_] := {Re[z], Im[z]}  
Manipulate[ParametricPlot[{z2xy[phi0[t]], z2xy[phi1[t]], z2xy[F[t, u]]},  
  {t, 0, 2 Pi}, PlotRange -> {{-3, 3}, {-3, 3}},  
  {u, 0, 1}]
```

(もっと面白い例に変えて、図もつけたいな…)

参考文献

[1]